
Interfacing MCP6S2X PGAs to PICmicro® Microcontroller
INTRODUCTION
The MCP6S21/2/6/8 family of one, two, six or eight
channel Programmable Gain Amplifiers (PGA) commu-
nicate using a standard 3-wire Serial Peripheral Inter-
face (SPI™) protocol. This application note shows how
to program the six channel MCP6S26 PGA gains,
channels and shutdown registers using the PIC16C505
microcontroller. 
The PIC16C505 microcontroller does not have a hard-
ware SPI module, therefore, a firmware SPI (Bit Bang)
method is used to program the PGA. The MCP6S2X
Evaluation Board was used to develop this application
note firmware. For additional information, refer to the
MCP6S2X Evaluation Board User’s Guide (DS51327),
MCP6S21/2/6/8 PGA data sheet (DS21117) and
PIC16C505 data sheet (DS40192).

COMMUNICATION
Figure 1 shows the SPI communication format with the
clock idle low and the serial data latched at the rising
edge of the clock. This format is sometimes referred to
as “0,0” mode. Instructions for the MCP6S21/2/6/8
family of PGAs consists of sixteen serial clock cycles
and two serial data bytes. The first byte is the instruc-
tion byte, which consists of register options, such as
Write, Shutdown, Gain and Channel. The second byte
is the data byte, whose three Least Significant Bits
(LSB) are used to program gain or channel. There are
eight gain settings and up to eight channel settings,
depending on the specific device.

Command bits 7, 6 and 5 of the instruction byte deter-
mines whether to write to the registers or shutdown the
device. A bit configuration of 010 directs the data byte
to be written to the Gain or Channel registers. When
writing to the registers, if command bit 0 is cleared, the
data byte is shifted into the Gain register. If the com-
mand bit 0 is set, the data byte is shifted into the
Channel register.
A 001 command bit configuration shuts down the
device. If shutdown is selected, command bit 0
becomes a “don’t care” bit.

Command bits 4, 3, 2 and 1, and data bits 7 through 3,
are not used and are “don’t cares”. Refer to Register 1,
Register 2 and Register 3 tables on pages 2 and 3.

FIGURE 1: PGA Instruction byte and data byte programming sequence.

REGISTER  1: INSTRUCTION REGISTER

REGISTER 2: GAIN REGISTER

W-0 W-0 W-0 U-x U-x U-x U-x W-0
M2 M1 M0 — — — — A0
bit 7 bit 0

bit 7-5 M2-M0: Command Bits
000 = NOP (Default) (Note 1)
001 = PGA enters Shutdown Mode as soon as a full 16-bit word is sent and CS is raised.

(Notes 1 and 2)
010 = Write to register.
011 = NOP (reserved for future use) (Note 1)
1XX = NOP (reserved for future use) (Note 1)

bit 4-1 Unimplemented: Read as ‘0’ (reserved for future use)
bit 0 A0: Indirect Address Bit

1 = Addresses the Channel Register
0 = Addresses the Gain Register (Default)

Note 1: All other bits in the 16-bit word (including A0) are don’t cares.

2: The device exits Shutdown mode when a valid command (other than NOP or Shut-
down) is sent and CS is raised; that valid command will be executed. Shutdown
does not toggle.

Legend:
R = Readable bit W = Writable bit U = Unimplemented bit, read as ‘0’
-n = Value at POR ’1’ = Bit is set ’0’ = Bit is cleared x = Bit is unknown

U-x U-x U-x U-x U-x W-0 W-0 W-0
— — — — — G2 G1 G0

bit 7 bit 0

bit 7-3 Unimplemented: Read as ‘0’ (reserved for future use)
bit 2-0 G2-G0: Gain Select Bits

000 = Gain of +1 (Default)
001 = Gain of +2
010 = Gain of +4
011 = Gain of +5
100 = Gain of +8
101 = Gain of +10
110 = Gain of +16
111 = Gain of +32

Legend:
R = Readable bit W = Writable bit U = Unimplemented bit, read as ‘0’
-n = Value at POR ’1’ = Bit is set ’0’ = Bit is cleared x = Bit is unknown

REGISTER  3: CHANNEL REGISTER

U-x U-x U-x U-x U-x W-0 W-0 W-0
— — — — — C2 C1 C0

bit 7 bit 0

bit 7-3 Unimplemented: Read as ‘0’ (reserved for future use)
bit 2-0 C2-C0: Channel Select Bits

MCP6S21
000 = CH0 (Default)
001 = CH0
001 = CH0
011 = CH0
100 = CH0
101 = CH0
110 = CH0
111 = CH0

MCP6S22
CH0 (Default)
CH1
CH0
CH1
CH0
CH1
CH0
CH1

MCP6S26
CH0 (Default)
CH1
CH2
CH3
CH4
CH5
CH0
CH0

MCP6S28
CH0 (Default)
CH1
CH2
CH3
CH4
CH5
CH6
CH7

Legend:
R = Readable bit W = Writable bit U = Unimplemented bit, read as ‘0’
-n = Value at POR ’1’ = Bit is set ’0’ = Bit is cleared x = Bit is unknown

IMPLEMENTATION
Microchip Technology Inc.’s PIC16C505 microcontrol-
ler is used to program the six channel MCP6S26 PGA
registers. Appendix A shows the assembly source
code used for this application note. This source code is
also availabe on the Microchip web site at www.micro-
chip.com. This code shows the communication
sequence necessary to program the PGA. 

The code structure is as follows: 
Initially, all the necessary constants, such as input/out-
put lines, instruction bytes, gains, channels and shut-
down are defined. Three Random Access Memory
(RAM) bytes are then reserved. These bytes are used
to temporarily store PGA register bits and SPI protocol
counter bits. Once the RAM location is reserved, the
code is originated at a program memory location and
the input/output ports are defined. 

This code has four short subroutines, labeled ‘gain’,
‘channel’, ‘shutdown’ and ‘bitbang’. These
subroutines are called using the ‘call’ instruction. 

The ‘gain’ and ‘channel’ Subroutines
The ‘gain’ and ‘channel’ subroutines send the
instruction and data bytes to program the PGA Gains
and Channel registers. These bytes are defined by the
user according to the constants defined within the
source code. These constants are also listed in
Table 1.

TABLE 1: SOURCE CODE CONSTANTS

The user can select the corresponding constant for the
gain and channel before the ‘call’ instructions (as
described within the source code) and compile the
source code. 

Before the ‘gain’ or the ‘channel’ subroutines are
called using the ’call’ instruction, the user-selected
constant is transferred into the RAM location, labeled
‘register’. Once these subroutines are called, chip-
select ‘cs’ (defined as the 4th line of port C) is pulled
low and the command byte (‘PrgGain’ for gain or
‘PgrChannel’ for channel) is loaded into the ‘w’ regis-
ter. The ‘call’ instruction for the ‘bitbang’ subroutine
is then executed to send the command bits to the PGA.
Next, data stored in ‘register’ RAM location is trans-
ferred to the ‘w’ register. The ‘bitbang’ subroutine is
called again to send the data to the PGA and,  after this
call, the chip-select line is pulled high.

The ‘shutdown’ Subroutine
The ‘shutdown’ subroutine sends the shutdown
instruction to the PGA. The shutdown ‘call’ is com-
mented ‘;’, therefore, this call is not executed. In order
to shutdown the PGA, the user must uncomment or
delete the comment ‘;’ and recompile the source code.
This will execute the ‘shutdown’ subroutine.
When the ‘shutdown’ subroutine is called, ‘cs’ is
pulled low and the command byte ‘PrgShdn’ is trans-
ferred into the ‘w’ register. The ‘bitbang’ subroutine is
then called to send the data. This subroutine is called
twice in order to complete the required 16 clock and
data cycles, as shown in Figure 1. The second byte
that is transferred to the PGA contains dummy bits.

The ‘bitbang’ Subroutine
The ‘bitbang’ subroutine transmits an eight bit serial
word to the PGA using a firmware SPI protocol (or bit-
bang). This subroutine sends out the data located in
the ‘w’ register. This register is transferred into a RAM
memory location, labeled ‘buffer’. A counter byte is
then loaded by hexadecimal ’08’ into RAM location
(labeled ‘SPIcounter’) to count the transferring bits.
The 7th bit of ‘buffer’ is transferred initially through
the data-out ‘do’ (defined as the 3rd line of port C) line.
Then the clock line ‘sck’ (defined as the 5th line of port
C) is pulsed and the ‘do’ line is cleared. The ‘buffer’
is then shifted to the left using the ‘rlf’ instruction in
order to send the next bit. This loop continues until the
counter byte is decremented to hexadecimal ’00’, indi-
cating that all bits are transferred.

CONCLUSION
This application note has illustrates a method of com-
munication with the MCP6S21/2/6/8 family of PGAs
using the PIC16C505 EPROM microcontroller.
Because the firmware can be applied to other families
of PICmicro microcontrollers, please refer to the
PIC16C505 data sheet for source code compatibilities.

Function Constants Bits

Gain 1 gain1 b’00000000’

Gain 2 gain2 b’00000001’

Gain 4 gain4 b’00000010’

Gain 5 gain5 b’00000011’

Gain 8 gain8 b’00000100’

Gain 10 gain10 b’00000101’

Gain 16 gain16 b’00000110’

Gain 32 gain32 b’00000111’

Channel 0 channel0 b’00000000’

Channel 1 channel1 b’00000001’

Channel 2 channel2 b’00000010’

Channel 3 channel3 b’00000011’

Channel 4 channel4 b’00000100’

Channel 5 channel5 b’00000101’

Gain register PrgGain b’01000000’

Channel register PrgChannel b’01000001’

Shutdown PrgShdn b’00100000’
DS00248A-page 4  2003 Microchip Technology Inc.



AN248

Software License Agreement
The software supplied herewith by Microchip Technology Incorporated (the “Company”) is intended and supplied to you, the Com-
pany’s customer, for use solely and exclusively with products manufactured by the Company.
The software is owned by the Company and/or its supplier, and is protected under applicable copyright laws. All rights are reserved.
Any use in violation of the foregoing restrictions may subject the user to criminal sanctions under applicable laws, as well as to civil
liability for the breach of the terms and conditions of this license.
THIS SOFTWARE IS PROVIDED IN AN “AS IS” CONDITION. NO WARRANTIES, WHETHER EXPRESS, IMPLIED OR STATU-
TORY, INCLUDING, BUT NOT LIMITED TO, IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICU-
LAR PURPOSE APPLY TO THIS SOFTWARE. THE COMPANY SHALL NOT, IN ANY CIRCUMSTANCES, BE LIABLE FOR
SPECIAL, INCIDENTAL OR CONSEQUENTIAL DAMAGES, FOR ANY REASON WHATSOEVER.
APPENDIX A: ASSEMBLY SOURCE CODE
;--------------------------------------------------------------------------
;
; This source code programs the PGA registers based on the gain and channel
; variabls that user selects. User must type the proper gain and channel
; according to the listed definitions and compile this source code.
;
; File name: pga2pic.asm
; Date: 02/19/03
; File Version: 1.00
;
; Programmer: PRO MATE® II device programmer
; File Required: PIC16C505.inc
;
; Author: Ezana Haile
; Company: Microchip Technology Inc.
;
;--------------------------------------------------------------------------

ERRORLEVEL -302
ERRORLEVEL -305

#include <p16C505.inc>

__CONFIG _MCLRE_OFF & _WDT_OFF & _IntRC_OSC_RB4EN

; definitions for “data”
#define gain1 b’00000000’ ; Gain of 1
#define gain2 b’00000001’ ; Gain of 2
#define gain4 b’00000010’ ; Gain of 4
#define gain5 b’00000011’ ; Gain of 5
#define gain8 b’00000100’ ; Gain of 8
#define gain10 b’00000101’ ; Gain of 10
#define gain16 b’00000110’ ; Gain of 16
#define gain32 b’00000111’ ; Gain of 32
#define channel0 b’00000000’ ; channel 0
#define channel1 b’00000001’ ; channel 1
#define channel2 b’00000010’ ; channel 2
#define channel3 b’00000011’ ; channel 3
#define channel4 b’00000100’ ; channel 4
#define channel5 b’00000101’ ; channel 5

; definitions for "instructions"
#define PrgGain b’01000000’ ; instruction for gain register
#define PrgChannel b’01000001’ ; instruction for channel register
#define PrgShdn b’00100000’ ; instruction to shutdown PGA

; definitions for i/o ports for SPI communication

#define cs portc, rc4 ; chip select
#define sck portc, rc5 ; clock
#define do portc, rc3 ; buffer out

; reserve memory byte
cblock 0X10

SPIcounter, buffer, register
endc

;============================================================
;========== PROGRAM ============================
;============================================================
PGA ; code name

org 0x00 ; origination address
movlw b’00000111’ ; set the I/O for port c
tris portc
movlw b’00111111’ ; set the I/O for port b
tris portb
bsf cs ; unselect the device
bcf do ; keep the dataout (do) low
bcf sck ; clear clock
clrf buffer ; clear buffer
clrw ; clear w register

;*********************************************************************
; User can select the GAIN from the definition table provided above
; and type that as shown on the next line: “movlw gainx“.
; where X is: 1, 2, 4, 5, 8, 10, 16, or 32
;*********************************************************************

movlw gain2 ; <==== Change this gain constant
movwf register ; move the data in RAM location
call Gain ; program gain

;*********************************************************************
; User can select the Channel from the definition table provided above
; and type that as shown on the next line: “movlw channelx“.
; where X is: 0, 1, 2, 3, 4, or 5
;*********************************************************************

movlw channel0 ; <==== Change this channel constant
movwf register ; move the data in RAM location
call Channel ; program channel

;*********************************************************************
; If user needs to shutdown the PGA then the comment “;” needs
; to be removed from the following line: “ ; call Shutdown “
;*********************************************************************
; call Shutdown ; <==== uncomment this line for shutdown

goto finish ; done

;----------------------------------------------
;---- Gain programming subroutine
;---------------------------------------------- 
Gain

bcf cs ; select PGA
movlw PrgGain ; get the gain programing instruction
call bitbang ; send the bits through SPI

movf register,w
call bitbang ; send it through spi 
bsf cs ; unselect the devices

retlw 0 ; return from call

;----------------------------------------------
;---- Channel programming subroutine
;----------------------------------------------
Channel

bcf cs ; select PGA
movlw PrgChannel ; get the Channel programing instruction
call bitbang ; send the bits through SPI

movf register,w
call bitbang ; send it through spi
bsf cs ; unselect the devices

retlw 0 ; return from call

;----------------------------------------------
;---- Shutdown programming subroutine
;----------------------------------------------
Shutdown

bcf cs ; select PGA
movlw PrgShdn ; get the Channel programing instruction
call bitbang ; send the bits through SPI
call bitbang ; send dummy bits
bsf cs ; unselect the devices

retlw 0 ; return from call

;----------------------------------------------
;---- Bit Bang SPI communication routine
;----------------------------------------------
bitbang

movwf buffer ; move the data into RAM
clrc
movlw h’08’ ; counter
movwf SPIcounter ; set the bit bang counter

send btfsc buffer, 7 ; see the 7th bit of the buffer
bsf do ; if the 7th bit is set, then set the do high
bsf sck ; set clock (pulse clock)
bcf sck ; clear clock
bcf do ; clear the data
rlf buffer,f ; roll the bits
decfsz SPIcounter, f ; check end of counter
goto send ; loop

retlw 0 ; return from call

;============================================================
;============================================================

finish goto finish
end

NOTES:

Note the following details of the code protection feature on Microchip devices:
• Microchip products meet the specification contained in their particular Microchip Data Sheet.

• Microchip believes that its family of products is one of the most secure families of its kind on the market today, when used in the 
intended manner and under normal conditions.

• There are dishonest and possibly illegal methods used to breach the code protection feature. All of these methods, to our 
knowledge, require using the Microchip products in a manner outside the operating specifications contained in Microchip's Data 
Sheets. Most likely, the person doing so is engaged in theft of intellectual property.

• Microchip is willing to work with the customer who is concerned about the integrity of their code.

• Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not 
mean that we are guaranteeing the product as “unbreakable.”

Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of our
products. Attempts to break microchip’s code protection feature may be a violation of the Digital Millennium Copyright Act. If such
acts allow unauthorized access to your software or other copyrighted work, you may have a right to sue for relief under that Act.
Information contained in this publication regarding device
applications and the like is intended through suggestion only
and may be superseded by updates. It is your responsibility to
ensure that your application meets with your specifications. No
representation or warranty is given and no liability is assumed
by Microchip Technology Incorporated with respect to the
accuracy or use of such information, or infringement of patents
or other intellectual property rights arising from such use or
otherwise. Use of Microchip’s products as critical components in
life support systems is not authorized except with express
written approval by Microchip. No licenses are conveyed,
implicitly or otherwise, under any intellectual property rights.
 2003 Microchip Technology Inc.

